package ch11
{
	import flash.display.Sprite;
	import flash.display.StageAlign;
	import flash.display.StageScaleMode;
	import flash.events.Event;
	import flash.geom.Point;

	import utils.Ball;

	/**
	 * 两个轴上的动量守恒.
	 * 多物体.
	 * 
	 * 解决潜在问题:
	 * 两个物体之间仍有可能在某些情况下粘在一起。
	 * 最有可能发生在影片非常拥挤的条件下，并且在运动速度很快时结果会更糟糕
	 * @author sjp
	 *
	 */
	public class MultiBilliard2 extends Sprite
	{
		private var balls:Array;
		private var numBalls:uint = 20;
		private var bounce:Number = -1.0;

		public function MultiBilliard2()
		{
			init();
		}

		private function init():void
		{
			stage.scaleMode = StageScaleMode.NO_SCALE;
			stage.align = StageAlign.TOP_LEFT;

			balls = new Array();
			for (var i:uint = 0; i < numBalls; i++)
			{
				var radius:Number = Math.random() * 50 + 20;
				var ball:Ball = new Ball(radius);
				ball.mass = radius;
				ball.x = Math.random() * stage.stageWidth;
				ball.y = Math.random() * stage.stageHeight;
				ball.vx = Math.random() * 10 - 5;
				ball.vy = Math.random() * 10 - 5;
				addChild(ball);
				balls.push(ball);
			}

			addEventListener(Event.ENTER_FRAME, onEnterFrame);
		}

		private function onEnterFrame(event:Event):void
		{
			for (var i:uint = 0; i < numBalls; i++)
			{
				var ball:Ball = balls[i];
				ball.x += ball.vx;
				ball.y += ball.vy;
				checkWalls(ball);
			}
			for (i = 0; i < numBalls - 1; i++)
			{
				var ballA:Ball = balls[i];
				for (var j:Number = i + 1; j < numBalls; j++)
				{
					var ballB:Ball = balls[j];
					checkCollision(ballA, ballB);
				}
			}
		}

		private function checkWalls(ball:Ball):void
		{
			if (ball.x + ball.radius > stage.stageWidth)
			{
				ball.x = stage.stageWidth - ball.radius;
				ball.vx *= bounce;
			}
			else if (ball.x - ball.radius < 0)
			{
				ball.x = ball.radius;
				ball.vx *= bounce;
			}
			if (ball.y + ball.radius > stage.stageHeight)
			{
				ball.y = stage.stageHeight - ball.radius;
				ball.vy *= bounce;
			}
			else if (ball.y - ball.radius < 0)
			{
				ball.y = ball.radius;
				ball.vy *= bounce;
			}
		}

		private function checkCollision(ball0:Ball, ball1:Ball):void
		{
			var dx:Number = ball1.x - ball0.x;
			var dy:Number = ball1.y - ball0.y;
			var dist:Number = Math.sqrt(dx * dx + dy * dy);
			if (dist < ball0.radius + ball1.radius)
			{
				// 计算角度和正余弦值 
				var angle:Number = Math.atan2(dy, dx);
				var sin:Number = Math.sin(angle);
				var cos:Number = Math.cos(angle);
				/**
				 * 旋转坐标至x轴水平，使得碰撞只与x轴有关，与y轴无关。
				 * 记住 vy 的值永远不变。
				 * */
				// 旋转 ball0 的位置 
				var pos0:Point = new Point(0, 0);
				// 旋转 ball1 的速度 
				var pos1:Point = rotate(dx, dy, sin, cos, true);
				// 旋转 ball0 的速度 
				var vel0:Point = rotate(ball0.vx, ball0.vy, sin, cos, true);
				// 旋转 ball1 的速度 
				var vel1:Point = rotate(ball1.vx, ball1.vy, sin, cos, true);
				// 碰撞的作用力
				var vxTotal:Number = vel0.x - vel1.x;
				vel0.x = ((ball0.mass - ball1.mass) * vel0.x + 2 * ball1.mass * vel1.x) / (ball0.mass + ball1.mass);
				vel1.x = vxTotal + vel0.x;

				/**解决两个物体黏在一起的问题.
				 * 首先确定绝对速度（自创的一个词），是所有速度的绝对值之和。
				 * 例如，如果一个速度是 -5 另一个速度是 10，那么绝对值就是 5 和 10，总和就是 5 + 10 = 15。 
				 * 接下来，确定小球之间重叠部分的大小。用总半径长度减去总距离。 
				 * 然后根据小球速度与绝对速度的百分比，让小球移动出重叠的那一部分。
				 *  结果会让小球之间没有重叠。这种方法比早前的版本要复杂一些，但确消除了很多 bug。
				 * */
				// 更新位置
				var absV:Number = Math.abs(vel0.x) + Math.abs(vel1.x);
				var overlap:Number = (ball0.radius + ball1.radius) - Math.abs(pos0.x - pos1.x);
				pos0.x += vel0.x / absV * overlap;
				pos1.x += vel1.x / absV * overlap;

				/**原有方法*/
				// 更新位置
//				pos0.x += vel0.x;
//				pos1.x += vel1.x;
				// 将位置旋转回来
				var pos0F:Object = rotate(pos0.x, pos0.y, sin, cos, false);
				var pos1F:Object = rotate(pos1.x, pos1.y, sin, cos, false);
				/**
				 * 最终的位置与系统中心点 ball0 的位置相对的。
				 * 因此，需要把它们都加到 ball0 的位置上，从而得到实际在屏幕上的位置。
				 * */
				// 将位置调整为屏幕的实际位置 
				ball1.x = ball0.x + pos1F.x;
				ball1.y = ball0.y + pos1F.y;
				ball0.x = ball0.x + pos0F.x;
				ball0.y = ball0.y + pos0F.y;
				// 将速度旋转回来 
				var vel0F:Object = rotate(vel0.x, vel0.y, sin, cos, false);
				var vel1F:Object = rotate(vel1.x, vel1.y, sin, cos, false);
				ball0.vx = vel0F.x;
				ball0.vy = vel0F.y;
				ball1.vx = vel1F.x;
				ball1.vy = vel1F.y;
			}
		}

		private function rotate(x:Number, y:Number, sin:Number, cos:Number, reverse:Boolean):Point
		{
			var result:Point = new Point();
			if (reverse)
			{
				result.x = x * cos + y * sin;
				result.y = y * cos - x * sin;
			}
			else
			{
				result.x = x * cos - y * sin;
				result.y = y * cos + x * sin;
			}
			return result;
		}
	}
}